********************************************************************************
** 	TITLE:		repprim2008_polls		                                      ** 	
**	AUTHOR:	    Philippe Mongrain                                             **
**  DATE:	    October 2022 						                          **	
**  VERSION:	Stata 16					                                  **	
********************************************************************************

* Version control

version 16.0

* Import data

import excel surveys, firstrow sheet(repprim2008) clear
destring _all, replace

* Generate survey date

gen date = date(polldate, "MDY")
format %tdMon_DD,_CCYY date
drop polldate
rename date polldate

* Generate election date

gen edate = 20080904

gen electiondate = date(string(edate,"%8.0f"),"YMD")

format %tdMon_DD,_CCYY electiondate

* Time of survey

gen time = (electiondate - polldate)

* Generate mean vote intention value by day

bysort time : egen giulianivote = mean(giuliani)
bysort time : egen huckabeevote = mean(huckabee)
bysort time : egen mccainvote = mean(mccain)
bysort time : egen paulvote = mean(paul)
bysort time : egen romneyvote = mean(romney)
bysort time : egen thompsonvote = mean(thompson)
bysort time : egen othersvote = mean(others)

sort time

* Drop duplicates

duplicates tag polldate, gen(dup)
duplicates drop polldate, force
drop dup

* Reshape the dataset

rename giuliani v_giuliani
rename huckabee v_huckabee
rename mccain v_mccain
rename paul v_paul
rename romney v_romney
rename thompson v_thompson
rename others v_others

reshape long v_, i(polldate) j(party) string

rename v_ vote

keep polldate electiondate party vote poll time

order poll polldate electiondate time party vote

* Generate rank of parties

gsort polldate -vote

bysort polldate : gen rank = _n

gen first = party if rank == 1
gen second = party if rank == 2
gen third = party if rank == 3

bysort polldate : gen winner = first[1]
bysort polldate : gen runnerup = second[2]
bysort polldate : gen thirdplace = third[3]

* Generate poll margin

bysort polldate : gen pollmar = vote[1] - vote[2]

* Drop duplicates

duplicates tag polldate, gen(dup)
duplicates drop polldate, force
drop dup

* Misleading poll

gen misleading = 0 if winner == "mccain" & pollmar >= 1
replace misleading = 1 if misleading!=0

* Save

drop if time == 0 | time == .

keep polldate pollmar misleading time

save "repprim2008_polls.dta", replace